

# GEOMETRY OF THE PRIMARY SURFACE OF THE GBT

F. R. SCHWAB

January 17, 1990

**Introduction.** In this memorandum I describe the idealized geometry of the primary reflecting surface of the Green Bank Telescope, emphasizing, in particular, the differential geometric aspects that might be pertinent to design and fabrication of the surface panels. A handy reference on the relevant mathematics is Dirk Struik's *Lectures on Classical Differential Geometry*, Second Edition, Addison-Wesley (1961), from which I've borrowed freely.

In the final few sections I present some numerical calculations that might help to establish some of the details of panel manufacture, in particular, the number of molds that might be required.

**The Design Paraboloid.** I assume that the primary reflector will be a portion of a paraboloid of revolution, whose profile  $z = f(r)$  is the parabola  $f(r) = \frac{r^2}{4c}$  of focal length  $c$ . Here,  $z$  represents the height of the surface above an  $x$ - $y$  plane which is tangent to the vertex of the paraboloid;  $r$  is defined as  $\sqrt{x^2 + y^2}$ ; and  $x$ ,  $y$ , and  $z$  are given in meters. The current design specification for the focal length is  $c = 60$  meters. The azimuthal coordinate is  $\varphi \equiv \tan^{-1} \frac{y}{x}$ .

**Surface Curvature.** The lines of extremal curvature are the radial and azimuthal lines, i.e., the meridians and parallels. At a point  $P$  on the surface, the normal curvature  $\kappa_1$  along the radial direction, or the meridian, through  $P = (x, y, f(r))$  is identical to the normal curvature of the surface profile  $f(r)$ ; i.e.,

$$\kappa_1 = \frac{4c^2}{(r^2 + 4c^2)^{3/2}}. \quad (1)$$

The radius of curvature in the radial direction is  $R_1 \equiv \kappa_1^{-1}$ . The normal curvature in the azimuthal direction is

$$\kappa_2 = \frac{1}{\sqrt{r^2 + 4c^2}}, \quad (2)$$

and the radius of curvature  $R_2$ , measured in that direction, is equal to  $\kappa_2^{-1}$ .  $R_2$  is the length of the normal to the surface profile, measured between  $P$  and the axis of revolution. At the vertex,  $R_1 = R_2 = 2c$ . Away from the vertex,  $\kappa_1$  is less than  $\kappa_2$ , and  $R_1$  exceeds  $R_2$ . Figure 1 shows a plot of the principal radii of curvature  $R_1$  and  $R_2$ , and Figure 2 shows the principal curvatures  $\kappa_1$  and  $\kappa_2$ .

The curvature in an arbitrary direction at any point on the surface can, according to a theorem of Euler, be expressed in terms of  $\kappa_1$  and  $\kappa_2$ : A curve passing

through the point  $P$ , and intersecting the radial line through  $P$  at an angle  $\alpha$ , has as its normal curvature at  $P$

$$\kappa = \kappa_1 \cos^2 \alpha + \kappa_2 \sin^2 \alpha. \quad (3)$$

Always, in our case,  $\kappa_1 \leq \kappa \leq \kappa_2$ . Equation 3 may be useful in answering questions on panel manufacture that arose at the recent GBT Design Review meeting (see sections further below). (Note that the angle  $\alpha$  is the actual angle between the two curves—not the angle of, say, the  $x$ - $y$  projection of the intersection of the curves.<sup>1</sup>)

The intersection of the surface with a plane close to the tangent plane and parallel to it is, to first approximation, similar to an ellipse called the *Dupin indicatrix*. This ellipse has principal semi-axes  $\sqrt{R_1}$  and  $\sqrt{R_2}$  which are oriented in the corresponding directions of principal curvature. Since  $R_1 \geq R_2$ , the surface panels, if they are elongated, should be elongated in the radial direction. An appropriate fractional degree of elongation would be  $\sqrt{\frac{R_1}{R_2}} = \frac{\sqrt{r^2 + 4c^2}}{2c}$ . For  $c = 60$ , this ratio increases from unity to approximately 1.323 as  $r$  varies between 0 and 104 meters.

The Gaussian, or intrinsic, curvature (sometimes called the total curvature) is given by  $K = \kappa_1 \kappa_2$ . The mean, or extrinsic, curvature is  $M = (\kappa_1 + \kappa_2)/2$ .

**Element of Length, and Element of Surface Area.** The square of the element of length on the surface is given by

$$\begin{aligned} ds^2 &= \left(1 + \frac{r^2}{4c^2}\right) dr^2 + r^2 d\varphi^2 \\ &= \left(1 + \frac{x^2}{4c^2}\right) dx^2 + \frac{xy}{2c^2} dx dy + \left(1 + \frac{y^2}{4c^2}\right) dy^2. \end{aligned} \quad (4)$$

The element of surface area is given by

$$dA = \sqrt{1 + \frac{x^2 + y^2}{4c^2}} dx dy = r \sqrt{1 + \frac{r^2}{4c^2}} dr d\varphi. \quad (5)$$

Numerical integration of this quantity, using the values which correspond to the current design specifications for the primary reflecting surface of the Green Bank Telescope—namely, taking  $c = 60$  and integrating over a disk in the  $x$ - $y$  plane of radius fifty meters, centered at a distance of 54 meters from the origin—yields a total surface area (to the nearest square centimeter) of 8893.1888 square meters.

When, as I do below, one calculates, say, the r.m.s. difference between the nominal design paraboloid and the nominal figure of the surface panels, it is appropriate to measure the needed distances in directions normal to the nominal design paraboloid. The integration should be weighted appropriately—that is, it should be with respect to the element of surface area expressed in terms of the chosen variables of integration, and it should be normalized by  $\iint dA$ , taken over the selected region.

---

<sup>1</sup>  $\alpha$  is related to the projected angle  $\theta$  according to  $\cos \alpha = \sqrt{\frac{r^2 + 4c^2}{r^2 \cos^2 \theta + 4c^2}} \cos \theta$ .

**Surface Normal.** The unit normal  $N$ , with positive  $z$ -component, at the point  $P = (x, y, f(r))$  is given by

$$N = \frac{(-x, -y, 2c)}{\sqrt{r^2 + 4c^2}}. \quad (6)$$

**Are Spherical Molds a Viable Possibility?** The calculations presented in this, and the following sections are based on a short Fortran program, a listing of which is shown in the Appendix. The program calculates the r.m.s. difference between the nominal design paraboloid of the Green Bank Telescope and the nominal figure of possible surface panels. In this section I shall investigate the use of spherical panels. That is, I shall assume here that each panel is a “cap” cut out of a sphere of appropriate radius. (The question of whether spherical panel shapes could be used arose at the December GBT Design Review Meeting.)

In Table 1 I show the results for spherical panels of diameter 2.5 meters. The radius of curvature  $R$  of each panel is assumed to be equal to the reciprocal of the mean curvature  $M = (\kappa_1 + \kappa_2)/2$  of the design paraboloid, measured at the position of the center of the panel.

**Table 1.** Results for Spherical Panels

Panel Location, $r$ (meters)	$R$ (meters)	R.m.s. Error (microns)
0	120.0	0.079
10	120.8	9.38
20	123.3	36.1
30	127.4	78.5
40	133.1	133
50	140.4	199
60	149.1	271
70	159.1	348
80	170.4	430
90	182.9	514
100	196.5	600
104	202.1	635

This calculation probably rules out the use of spherical molds, since panels of 100 micron r.m.s. accuracy are what I believe are called for in the current error budget.<sup>2</sup> I do, however, assume that the sphere is centered on the surface normal passing through the central point of the region of interest, at a distance  $R$  away from that point; therefore, by moving the sphere slightly closer to the paraboloid, the r.m.s. error could be approximately halved.

<sup>2</sup>Although, if I repeat this calculation using a panel diameter of one meter, I find that the r.m.s. error at  $r = 104$  meters is reduced to 102 microns.

Figure 3 shows the principal radii of curvature, out to  $r = 5$  meters, for a ten-meter diameter paraboloidal reflector of  $f/D = 1.75$ , corresponding to the size and the focal ratio of the Keck Telescope. For that case, the principal radii of curvature differ by only a bit more than two percent at the edge of the reflector. In the case of the Green Bank Telescope, the principal radii of curvature at the far edge of the reflector differ by about 75 percent (at  $r = 104$  m,  $R_1 \approx 278$  m and  $R_2 \approx 159$  m); this is why reasonably-sized spherical panels would not fulfill our requirements.

**How Many Molds Are Required?** Precision, non-spherical molds seem to be required. In this section I attempt to calculate the least number of precision molds that would be required for the Green Bank Telescope. For these calculations, I have again used the computer program shown in the Appendix. For simplicity, the program uses for the nominal figure of the surface panel not an actual segment of the design paraboloid—but rather a figure having a circular profile about every angle through a central point, with an ‘appropriate’ radius of curvature at each of these angles. This radius of curvature is ‘appropriate’ in the sense that it matches the radius of curvature of the design paraboloid, at each of these angles, at some fixed location on the surface. In other words, these radii of curvature vary between the two principal radii of curvature (appropriate to the design location of the panel) in accord with Equation 3.

Results of an illustrative calculation are summarized in Table 2, where, like before, I have assumed a panel diameter of 2.5 meters. A discrete set of panel radii is given in the first column of the table; the radial distance of the location on the surface to which the panel is most precisely cut is given in the second column, labeled ‘Design Radius’; the principal radii of curvature at that design location are given in the third and fourth columns; and the r.m.s. error at the given radius is shown in the fifth column. According to this calculation, approximately eleven molds suffice to achieve r.m.s. errors less than 100 microns at each of the panel locations.

This is a conservative calculation, in that all panels have an r.m.s. error less than the overall r.m.s. accuracy that is desired of the surface. And, as before, the r.m.s. errors could be further reduced by re-distributing the errors more equally behind, and in front of, the design paraboloid.

On the other hand, we have been discussing the use of trapezoidal panels of maximum *side* 2.5 meters, which would not quite fit within the disks used in this calculation. Thus, this sample calculation, though it may appear promising, is probably a bit too rough and preliminary. My program will need to be double-checked for validity, by myself and others in the GBT design group, before these calculations are taken as ‘Gospel’.

**Table 2.** Possible Utilization of Precision Molds

Panel Location, $r$ (meters)	Design Radius (meters)	$R_1$ (meters)	$R_2$ (meters)	R.m.s. Error (microns)
5.25	17.75	124.0	121.3	77.7
7.75				68.9
10.25				56.7
12.75				41.1
15.25				22.3
17.75				3.4
20.25				25.8
22.75				54.4
25.25				86.2
27.75	32.75	133.7	124.4	76.4
30.25				39.9
32.75				5.9
35.25				43.0
37.75				88.0
40.25	45.25	146.5	128.2	99.4
42.75				51.3
45.25				7.4
47.75				53.9
50.25	52.75	156.4	131.1	56.5
52.75				8.1
55.25				58.9
57.75	60.25	168.1	134.3	60.6
60.25				8.6
62.75				62.7
65.25	67.75	181.7	137.8	63.5
67.75				9.0
70.25				65.4
72.75	75.25	197.3	141.6	65.5
75.25				9.2
77.75				67.1
80.25	82.75	215.1	145.8	66.6
82.75				9.3
85.25				68.0
87.75	90.25	235.1	150.2	67.0
90.25				9.3
92.75				68.1
95.25	97.75	257.5	154.8	66.7
97.75				9.2
100.25				67.7
102.75	102.75	273.8	158.0	9.1




Figure 1.




Figure 2.




Figure 3.



# Appendix

```

program patchcalc
c This program calculates the r.m.s. difference between the nominal
c design paraboloid of the GBT and the nominal figure of its surface
c panels. The panel may be cut to match a different location on the
c surface than one at which the calculation is carried out. For
c simplicity, the nominal figure of the surface panel is taken not to be
c an actual segment of the design paraboloid, but rather to have a circular
c profile about every angle through the center and to have the appropriate
c radius of curvature at each of these angles. (I.e., these panels
c could be cut by a fixed-height pendulum bob swinging over a rotating
c mold blank, with the length of the pendulum arm varying appropriately
c with the rotation.) The radii may be set equal by modifying two
c lines of the program, in case that one wishes to investigate the use of
c spherical panels.
    implicit real*8 (a-h,o-z)
    common rcent,rdesign,r1,r2,c
    external f,g,h,fn
c
c Main parameters:
c
c   c: focal length of the design paraboloid, in meters, (fixed at c=60).
c   rmax: panel radius (for simplicity, consider circular patches,
c         even though paraboloid can't be tiled by circular patches);
c         rmax=1.25 meters, for example.
c   rcent: radial distance (sqrt(x^2+y^2)) of panel center; i.e.,
c         something in the range between 4 and 104 meters.
c   rdesign: radial distance that panel is cut (designed) to match;
c         i.e., something in the range between 4 and 104 meters.
c   c=60d0
c Read rmax:
    print *, 'Type rmax'
    accept *, rmax
c Read panel center location and place that panel is designed for:
1   print *, 'Type rdesign,rcent'
    accept *, rdesign,rcent
c
c
c Calculate radii, r1 and r2, of principal curvature
c at place that panel is designed for:
    r1=(rdesign**2+4d0*c**2)**1.5d0/(4d0*c**2)
    r2=sqrt(rdesign**2+4d0*c**2)
c For panels cut from spheroid, use the following statements
c (use as the radius of curvature the reciprocal, 1/M, of the mean
c curvature, M, at the design radius):
    r1=1d0/(.5d0*(1d0/r1+1d0/r2))
    r2=r1
c
    print *, 'R1,R2=',r1,r2
c Now integrate the squared distance between the two surfaces
c (measured normal to the design paraboloid), using an IMSL routine,
c dtwodq, to do the numerical quadrature:
    pi=4d0*atan(1d0)
    errabs=1d-12
    errrel=1d-6
    irule=2
c The first call integrates the square of the distance between the
c design paraboloid and panel, measured normal to the design
c paraboloid. The squared distance is weighted by the appropriate
c element of surface area, expressed in terms of the variables of
c integration.

```

```

        call dtwodq(f,Od0,rmax,g,h,errabs,errrel,irule,result,err)
        print *,result,err
c The second call integrates the element of surface area over the
c desired region. This is needed in order to calculate a properly
c normalized r.m.s. error.
        call dtwodq(fn,Od0,rmax,g,h,errabs,errrel,irule,result2,err)
c Calculate and print out the r.m.s. error:
        sigma=sqrt(result/result2)*1e6
        print *,'rms error (in microns)=' ,sigma
c Repeat the calculation for another pair, (rcent,rdesign):
        go to 1
    end

    double precision function f(smallr,theta)
c This function subroutine calculates the square of the distance
c between the design paraboloid and the panel patch, measured normal
c to the design paraboloid, and weighted by the element of surface area,
c the latter calculated appropriately for the chosen variables of
c integration.
        implicit real*8 (a-h,o-z)
        common rcent,rdesign,r1,r2,c
c Location of the center of the region of interest:
        xc=rcent
        yc=Od0
        zc=(xc**2+yc**2)/(4d0*c)
c Calculate the radius of curvature in the direction theta.
c (theta is the direction in x-y projection; alpha is the corresponding
c angle between curves on the surface that pass through the panel
c center and have projected angle theta.)
        xxx=(rdesign**2+4d0*c**2)/((rdesign*cos(theta))**2+4d0*c**2)
        alpha=acos(sqrt(xxx)*cos(theta))
        rcurv=1d0/(cos(alpha)**2/r1+sin(alpha)**2/r2)
c Calculate the center of the osculating circle:
        x0=xc-rcurv*xc/sqrt(xc**2+yc**2+4d0*c**2)
        y0=yc-rcurv*yc/sqrt(xc**2+yc**2+4d0*c**2)
        z0=zc+rcurv*2d0*c/sqrt(xc**2+yc**2+4d0*c**2)
c Calculate the x-y coordinates of the point of interest on the design
c paraboloid:
        x=xc+smallr*cos(theta)
        y=yc+smallr*sin(theta)
        R=rcurv
c Finally, T1 is the distance between the design paraboloid and the panel
c patch, measured normal to the design paraboloid. (This statement was
c generated by the algebraic manipulation program, MACSYMA. I haven't
c bothered to simplify the equation it represents.):
        T1=(-SQRT(Y**2+X**2+4*C**2)*SQRT(C**2*(-16*Y**2*Z0**2+X**2*(-16*Z
1      0**2-16*Y0**2+16*Y*Y0-32*Y**2)+X*X0*(32*Y*Y0+16*Y**2)+16*Y**3*Y
2      0-16*Y**4+R**2*(16*Y**2+16*X**2)-16*X0**2*Y**2+16*X**3*X0-16*X*
3      *4)+C**3*(-64*Y*Y0*Z0+64*Y**2*Z0-64*X*X0*Z0+64*X**2*Z0)+C*(8*Y*
4      *4*Z0+16*X**2*Y**2*Z0+8*X**4*Z0)+C**4*(-64*Y0**2+128*Y*Y0-64*Y*
5      *2-64*X0**2+128*X*X0-64*X**2+64*R**2)-Y**6-3*X**2*Y**4-3*X**4*Y
6      **2-X**6)+8*C**2*SQRT(Y**2+X**2+4*C**2)*Z0+C*SQRT(Y**2+X**2+4*C
7      **2)*(-4*Y*Y0+2*Y**2-4*X*X0+2*X**2))/(C*(4*Y**2+4*X**2)+16*C**3
8      )
c Square the normal distance, and weight by the element of surface
c area:
        wt=smallr*sqrt(1d0+(x**2+y**2)/4d0/c**2)
        f=wt*t1**2
        return
    end

```

```

      double precision function g(x)
c   Function subroutines g and h simply set the limits of integration,
c   in the azimuthal direction --- 0 and 2pi --- for the numerical
c   quadrature routine dtwodq.
      implicit real*8 (a-h,o-z)
      g=0d0
      return
      end
      double precision function h(x)
      implicit real*8 (a-h,o-z)
      h=8d0*atan(1d0)
      return
      end

      double precision function fn(smallr,theta)
c   This function subroutine returns the element of surface area,
c   expressed in terms of the chosen variables of integration.
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,r1,r2,c
      x=rcent+smallr*cos(theta)
      y=smallr*sin(theta)
      wt=smallr*sqrt(1d0+(x**2+y**2)/4d0/c**2)
      fn=wt
      return
      end

```